

Student Placement: Using SAS® To Combine and Prioritize Information

Smith, Kelly D.; Earls, Paul G.; Frye, Bobbie E., Central Piedmont Community College

ABSTRACT

When a student applies to a North Carolina community college in order to take curriculum classes requiring college-level proficiency, it is still common for placement tests to be administered so the college-readiness of the student can be assessed. However, placement tests are not the only mechanism used to determine a student's final placement. Other considerations include state initiatives such as Multiple Measures, transfer credit from other institutions of higher education, and overrides provided by community college advisors. In a recent effort to standardize and improve student placement data, the Office of Institutional Research at Central Piedmont Community College (CPCC) revised the SAS code used to finalize student placement.

This paper summarizes the refinement of the student placement program. Final student placement is constructed from four data sets: (a) the testing file, which is based on information from the student test data files; (b) the curriculum students by term file, which contains all students who participated in the registration process; (c) the transfer credit file, which contains all transfer credits awarded to students for a particular academic year; and (d) the overrides file, which contains all overrides provided to students who participated in the academic term. The correct determination of student placement requires the precise selection of appropriate information from the transfer and override data sets with WHERE, the careful use of MERGE procedures to combine data sets, the creation of flag variables to track merged information, and the meticulous application of IF THEN processing. The refinement of the SAS code for determining student placement illustrates the logic analysis required to construct SAS programs that generate accurate outcomes.

INTRODUCTION

Accurate information on academic placement levels of beginning students is a necessary component of student outcome analysis for the Voluntary Framework of Accountability (VFA). Since developing our initial placement program for VFA, we have increasingly used placement information for internal requests and research projects. The initial placement program imported comma separated values (CSV) files with override information and utilized a series of tracking variables to determine placement according to VFA guidelines (Smith, Frye, & Earls, 2017).

As a first step in improving our program, we decided to create data sets of override information (Smith & Earls, 2018). We also decided to refine the logic process of placement determination as another means of improving the placement program. The original program used tracking variables to determine placement in an ongoing process as additional data was pulled into the program. In the updated program, flag variables hold placement information until all data has been pulled in. Final placement determinations are made as the last step in the programming process. Waiting until the end to make changes to a student's placement enables us to better track how transfer credits and overrides are impacting student placement assignments.

INITIAL PLACEMENT

Initial placement of new students at CPCC is based on test scores and a recent state initiative entitled Multiple Measures. The Multiple Measures initiative is aimed at recent high school graduates and provides a means of obtaining college level placement based on a combination of high school GPA and high school coursework. Figure 1 offers an overview of the Multiple Measure process (gateway math and English equivalent to college level placement):




Figure 1. Overview of Multiple Measures

The testing file contains data on student test scores and Multiple Measures designations (provided by the North Carolina Department of Public Instruction) for the majority of students who participate in the registration process for the designated term. The testing file includes information on the English and math course assignments for students, and a placement variable that provides a shorthand designation for Multiple Measures. Each course assigned to a student starts with a two digit number, which is used in the following code to assist in the initial placement determination:

```

IF Placement EQ 2 THEN EngPlace='Multiple Measures';
ELSE IF SUBSTR(EngClass,1,2) IN
('02','03','04','05','06','08','12','13','14','15')
THEN EngPlace='Dev. Education';
ELSE IF EngClass EQ '07-College Level'
THEN EngPlace='College Level';
ELSE EngPlace='Not Determined';

IF Placement EQ 2 THEN MatPlace='Multiple Measures';
ELSE IF SUBSTR(MatClass,1,2) IN
('02','03','04','05','06','10','20','30','40','50','60','70','80')
THEN MatPlace='Dev. Education';
ELSE IF MatClass EQ '07-College Level'
THEN MatPlace='College Level';
ELSE MatPlace='Not Determined';

PROC SORT DATA=PlaceEngMath; BY ADCPersonID DESCENDING Placement
DESCENDING Score;
PROC SORT NODUPKEY DATA=PlaceEngMath OUT=NoDupPlaceEM;
BY ADCPersonID;

```

A student may appear more than once in the testing file; therefore, it is necessary to sort the information so the highest placement rises to the top before reducing the data set to one observation per student. The

MERGE procedure is used to combine the resulting data set with a list of all students who participated in registration for curriculum courses for the designated term (creating Place00 data set). If a student has no placement information in the merged data set, IF THEN is used to code their English/math placement as "Not Determined." As a tracking mechanism, the PROC FREQ procedure is used to determine how many students fall in each placement category:

```
PROC FREQ DATA=Place00;
TITLE "Placement based on Testing";
TABLES EngPlace MatPlace / NOROW NOCOL MISSING;
RUN;
```

TRANSFER CREDITS

One way a student's initial placement can be upgraded is through the use of transfer credit(s) from another institution of higher education. If a student has been successful in college-level English or math courses at another institution, their initial placement at CPCC can be modified to reflect transfer credit(s) from their prior institution. Information on transfer credits is already collected and maintained as a SAS data set at CPCC. The transfer credit file contains transfer credits for a wide variety of subjects in addition to math or English. Since our interest is only in gateway college-level math and English courses, we use a WHERE statement rather than a subsetting IF to pull in information in the most efficient manner possible:

```
WHERE SUBSTR(ADCXfrCourse,1,5) IN ('ENG-1', 'MAT-1')
AND ADCXfrCourse NOT IN ('MAT-101', 'MAT-102', 'ENG-101');
```

There are currently three introductory math/English courses offered that do not qualify as college-level: the second phrase in the WHERE statement ensures those classes are not pulled in for consideration. A two-step process of classifying the type of transfer credit follows. In the first step, transfer credit is identified as either being math or English:

```
IF SUBSTR(ADCXfrCourse,1,5)='ENG-1' THEN TransCL='E';
ELSE IF SUBSTR(ADCXfrCourse,1,5)='MAT-1' THEN TransCL='M';
ELSE DELETE;
PROC SORT NODUPKEY DATA=TransCL; BY ADCPersonID TransCL;
```

In the second step, two transfer flag variables and two data sets are created (math transfers, English transfers).

```
IF TransCL='E' THEN TransCLEng='Yes';
IF TransCL='M' THEN TransCLMat='Yes';

IF TransCLEng='Yes' THEN OUTPUT TransCLEng;
ELSE IF TransCLMat='Yes' THEN OUTPUT TransCLMath;
```

These data sets are then merged with the initial placement data set (Place00) to add the transfer flag variables in for later use. At this time, no changes to placement information are made:

```
DATA TransPlaceEng;
MERGE Place00 (IN=A)
TransCLEng (IN=B);
BY ADCPersonID;
IF A;
RUN;

DATA TransPlaceEngMath;
MERGE TransPlaceEng (IN=A)
TransCLMath (IN=B);
BY ADCPersonID;
```

```

    IF A;
    IF TransCLEng NE 'Yes' THEN TransCLEng='No';
    IF TransCLMat NE 'Yes' THEN TransCLMat='No';
RUN;

```

Before proceeding, PROC FREQ is used to determine how many students have transfer credit:

```

PROC FREQ DATA=TransPlaceEngMath;
TITLE "Students with English/Math Transfer Coursework";
TABLES TransCLEng TransCLMat / NOROW NOCOL MISSING;
RUN;

```

OVERRIDES

Initial student placement can also be modified through the use of advisor overrides during the registration process. Pulling in override information and creating override flag variables follows the same process as transfer credits:

```

WHERE SUBSTR(ADCStuOvrCourse,1,5) IN ('ENG-1','MAT-1')
      AND ADCStuOvrCourse NOT IN ('MAT-101','MAT-102','ENG-101');

IF SUBSTR(ADCStuOvrCourse,1,5)='ENG-1' THEN OverCL='E';
ELSE IF SUBSTR(ADCStuOvrCourse,1,5)='MAT-1' THEN OverCL='M';
ELSE DELETE;
PROC SORT NODUPKEY DATA=OverCL; BY ADCTPersonID OverCL;

IF OverCL='E' THEN OverCLEng='Yes';
ELSE IF OverCL='M' THEN OverCLMat='Yes';
IF OverCLEng='Yes' THEN OUTPUT OverCLEng;
ELSE IF OverCLMat='Yes' THEN OUTPUT OverCLMat;

```

Once the override flag variables and resulting data sets (math overrides, English overrides) have been created, they are merged with TransPlaceEngMath. Once again, no changes are made to students' initial placement:

```

DATA OvrTrnsPlcEng;
MERGE TransPlaceEngMath (IN=A)
      OverCLEng          (IN=B);
BY ADCTPersonID;
IF A;
RUN;

DATA OvrTrnsPlcEngMath;
MERGE OvrTrnsPlcEng (IN=A)
      OverCLMat      (IN=B);
BY ADCTPersonID;
IF A;
IF OverCLEng NE 'Yes' THEN OverCLEng='No';
IF OverCLMat NE 'Yes' THEN OverCLMat='No';
RUN;

```

PROC FREQ is used to determine the number of students with math/English overrides:

```

PROC FREQ DATA=OvrTrnsPlcEngMath;
TITLE "Students with English/Math Overrides";
TABLES OverCLEng OverCLMat / NOROW NOCOL MISSING;
RUN;

```

FINAL PLACEMENT

The data set, OvrTrnsPlcEngMath, contains flag variables for transfer credit and overrides in addition to students' initial math and English placement. From the PROC FREQ results, we also know the number of students in each placement category, how many students have transfer credits, and how many students have overrides.

To determine students' final math and English placements, we use a series of IF THEN statements. Final placement values reflect the impact of transfer credit and overrides. The statements are carefully ordered to reflect the higher priority placed on transfer credits as compared to overrides, and a final PROC FREQ is used so the impact of transfer credits and overrides on placement can be assessed (Table 1):

```
DATA FinalPlacement;
    SET OvrTrnsPlcEngMath;

    IF EngPlace NOT IN ("College Level", "Multiple Measures")
        AND TransCLEng EQ "Yes" THEN EngPlace="College Level by Transfer";

    ELSE IF EngPlace NOT IN ("College Level", "Multiple Measures",
                            "College Level by Transfer")
        AND OverCLEng EQ "Yes" THEN EngPlace="College Level by Override";

    ELSE EngPlace=EngPlace;

    IF MatPlace NOT IN ("College Level", "Multiple Measures")
        AND TransCLMat EQ "Yes" THEN MatPlace="College Level by Transfer";

    ELSE IF MatPlace NOT IN ("College Level", "Multiple Measures",
                            "College Level by Transfer")
        AND OverCLMat EQ "Yes" THEN MatPlace="College Level by Override";

    ELSE MatPlace=MatPlace;
RUN;

PROC FREQ DATA=FinalPlace&yt01;
    TITLE "Final Placement w/ Transfer Credit & Overrides Applied";
    TABLES EngPlace MatPlace / NOROW NOCOL;
RUN;
```

English Placement	Based on Testing	Final	Change
College Level	3,066	3,066	<i>n/a</i>
College Level by Override	<i>n/a</i>	448	+448
College Level by Transfer	<i>n/a</i>	452	+452
Developmental Education	2,169	2,045	-124
Multiple Measures	3,739	3,739	<i>n/a</i>
Not Determined	10,517	9,741	-776

Table 1. Impact of Transfer Credits and Overrides on English Placement

CONCLUSION

Accurate determination of student placement is essential for many reasons but it is rarely a straight forward process. Our initial student placement program was developed specifically for VFA reporting. Once we realized we were using placement information in new and unexpected ways, we decided to refine the program to provide more information. Through the use of selective data intake, the use of flag variables, and prioritized IF THEN processing, we have expanded the utility and quality of student placement information.

REFERENCES

Smith, K.D., Frye, B. E., & Earls, P. G. (2017). SAS® and the Voluntary Framework of Accountability: A prime example of the use of SAS in education. *SESUG 2017: The Proceedings of the 25th Annual SouthEast SAS Users Group, Cary, North Carolina*. Available at:
https://analytics.ncsu.edu/sesug/2017/SESUG2017_Paper-96_Final_PDF.pdf

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Kelly D. Smith
Central Piedmont Community College
704-330-2722 x3810
kelly.smith@cpcc.edu

Bobbie E. Frye
Central Piedmont Community College
704-330-6459
bobbie.frye@cpcc.edu

Paul G. Earls
Central Piedmont Community College
704-330-6399
paul.earls@cpcc.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.